Point symmetries of y''=0

Run TMF064.Package.m first!

```
Clear["Global`*"]
```

Variables and differential equations in the form R(x,u,∂u,...) = 0

```
(* Independent variables *)
IndepVar = {x};
(* Dependent variables *)
DepVar = {y};
(* PDEs, only the functions R(...) without "== 0" *)
PDEs = {D[y[x], x, x]}
```

```
{y''[x]}
```

Expression to substitute for in the infinitesimal criterion of invariance

```
subs = {D[y[x], x, x]};
sol = Solve[PDEs == 0, subs]
```

```
{{y''[x] → 0}}
```

Finding point symmetries by using a more and more specific ansatz

General ansatz

```
(* Infinitesimals for all variables *)
ξ[x] = Ξ[x, y[x]];
η[y] = H[x, y[x]];
(* Next expression should return zeroes
 if infinitesimals give a point symmetry of PDEs *)
zero = CheckPointSymmetryOfDE[PDEs, subs, IndepVar, DepVar, ξ, η]
```

```
y[0] = y[x]

y[1] = y'[x]

y[2] = y''[x]
```

```
{y[1]^2 H^(0,2)[x, y[0]] - y[1]^3 Ξ^(0,2)[x, y[0]] + 2 y[1] H^(1,1)[x, y[0]] -
  2 y[1]^2 Ξ^(1,1)[x, y[0]] + H^(2,0)[x, y[0]] - y[1] Ξ^(2,0)[x, y[0]]}
```

```
(* Coefficients of the polynomial in y[1] should be zero *)
Column[GetConditionsForPointSymmetries[zero, {y[1]}]]
```

$-\Xi^{(0,2)}[x, y[0]]$
$H^{(0,2)}[x, y[0]] - 2\,\Xi^{(1,1)}[x, y[0]]$
$H^{(2,0)}[x, y[0]]$
$2\,H^{(1,1)}[x, y[0]] - \Xi^{(2,0)}[x, y[0]]$

Second ansatz

```
ξ[x] = c[1] + c[2] x + c[3] x^2 + c[4] y[x] + c[5] x y[x] + c[6] x^2 y[x];
η[y] = d[1] + d[2] x + d[3] y[x] + d[4] x y[x] + d[5] y[x]^2 + d[6] x y[x]^2;
zero = CheckPointSymmetryOfDE[PDEs, subs, IndepVar, DepVar, ξ, η]
```

y[0] = y[x]

y[1] = y′[x]

y[2] = y″[x]

{-2 y[1] (c[3] - d[4] + c[6] y[0] -
2 d[6] y[0] + c[5] y[1] + 2 x c[6] y[1] - d[5] y[1] - x d[6] y[1])}

```
Column[GetConditionsForPointSymmetries[zero, {y[1], y[0], x}]]
```

-2 (c[3] - d[4])
-2 (c[5] - d[5])
-2 (c[6] - 2 d[6])
-2 (2 c[6] - d[6])

The last ansatz

```
nc = 8;
ξ[x] = c[1] + c[3] x + c[7] x^2 + c[5] y[x] + c[8] x y[x];
η[y] = c[2] + c[6] x + c[4] y[x] + c[7] x y[x] + c[8] y[x]^2;
zero = CheckPointSymmetryOfDE[PDEs, subs, IndepVar, DepVar, ξ, η]
```

y[0] = y[x]

y[1] = y′[x]

y[2] = y″[x]

{0}

Infinitesimal generators, point transformations and commutator table from the last ansatz

```
ShowPointSymmetriesAndCommutationRelations[X, f, t, IndepVar, DepVar, ξ, η, c, nc, {}]
```

Infinitesimal operators:

$X[1] f[x, y] = f^{(1,0)}[x, y]$

$X[2] f[x, y] = f^{(0,1)}[x, y]$

$X[3] f[x, y] = x f^{(1,0)}[x, y]$

$X[4] f[x, y] = y f^{(0,1)}[x, y]$

$X[5] f[x, y] = y f^{(1,0)}[x, y]$

$X[6] f[x, y] = x f^{(0,1)}[x, y]$

$X[7] f[x, y] = x y f^{(0,1)}[x, y] + x^2 f^{(1,0)}[x, y]$

$X[8] f[x, y] = y^2 f^{(0,1)}[x, y] + x y f^{(1,0)}[x, y]$

Corresponding global transformations:

X[1] gives $\{x[t] \to x + t, y[t] \to y\}$

X[2] gives $\{x[t] \to x, y[t] \to y + t\}$

X[3] gives $\{x[t] \to x e^t, y[t] \to y\}$

X[4] gives $\{x[t] \to x, y[t] \to y e^t\}$

X[5] gives $\{x[t] \to x + y t, y[t] \to y\}$

X[6] gives $\{x[t] \to x, y[t] \to y + x t\}$

X[7] gives $\left\{x[t] \to \frac{x}{1 - x t}, y[t] \to \frac{y}{1 - x t}\right\}$

X[8] gives $\left\{y[t] \to \frac{y}{1 - y t}, x[t] \to \frac{x}{1 - y t}\right\}$

Commutator table:

	1	2	3	4	5	6	7
1	0	0	X[1]	0	0	X[2]	2 X[3] + X
2	0	0	0	X[2]	X[1]	0	X[6]
3	-X[1]	0	0	0	-X[5]	X[6]	X[7]
4	0	-X[2]	0	0	X[5]	-X[6]	0
5	0	-X[1]	X[5]	-X[5]	0	-X[3] + X[4]	X[8]
6	-X[2]	0	-X[6]	X[6]	X[3] - X[4]	0	0
7	-2 X[3] - X[4]	-X[6]	-X[7]	0	-X[8]	0	0
8	-X[5]	-X[3] - 2 X[4]	0	-X[8]	0	-X[7]	0

Vector fields

```
xmin = -5.0; xmax = 5.0; ymin = -5.0; ymax = 5.0;
```

Translation

```
{VectorPlot[{1, 0}, {x, xmin, xmax}, {y, ymin, ymax}, ImageSize → Medium],
 VectorPlot[{0, 1}, {x, xmin, xmax}, {y, ymin, ymax}, ImageSize → Medium]}
```

{  ,

 }

Scaling

```
{VectorPlot[{x, 0}, {x, xmin, xmax}, {y, ymin, ymax}, ImageSize → Medium],
 VectorPlot[{0, y}, {x, xmin, xmax}, {y, ymin, ymax}, ImageSize → Medium]}
```

{



,



}

Rotation and Lorentz transformation

```
{VectorPlot[{-y, x}, {x, xmin, xmax}, {y, ymin, ymax}, ImageSize → Medium],
 VectorPlot[{-y, -x}, {x, xmin, xmax}, {y, ymin, ymax}, ImageSize → Medium]}
```

{ 

,



}

Galileo transformation

```
{VectorPlot[{0, x}, {x, xmin, xmax}, {y, ymin, ymax}, ImageSize → Medium],
 VectorPlot[{y, 0}, {x, xmin, xmax}, {y, ymin, ymax}, ImageSize → Medium]}
```





Projective transformation

```
{VectorPlot[{x^2, x y}, {x, xmin, xmax}, {y, ymin, ymax}, ImageSize → Medium],
 VectorPlot[{y x, y^2}, {x, xmin, xmax}, {y, ymin, ymax}, ImageSize → Medium]}
```

{  ,

 }

All transformations at once

```
Manipulate[VectorPlot[{Tx + Sx x + Gx y + Px x^2 + Py x y, Ty + Gy x + Sy y + Px x y + Py y^2},
   {x, xmin, xmax}, {y, ymin, ymax}],
 {{Tx, 0, "Translation in x"}, -10, 10, 1}, {{Sx, 0, "Scaling of x"}, -10, 10, 1},
 {{Gx, 0, "Galileo transformation in x"}, -10, 10, 1},
 {{Px, 0, "Projective transformation in x"}, -10, 10, 1},
 {{Ty, 0, "Translation in y"}, -10, 10, 1}, {{Sy, 0, "Scaling of y"}, -10, 10, 1},
 {{Gy, 0, "Galileo transformation in y"}, -10, 10, 1},
 {{Py, 0, "Projective transformation in y"}, -10, 10, 1}]
```